

Christoffel Symbols and Geodesic Equation

This is a *Mathematica* program to compute the Christoffel and the geodesic equations, starting from a given metric $g_{\alpha\beta}$. The Christoffel symbols are calculated from the formula

$$\Gamma^{\lambda}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma} (\partial_{\mu} g_{\sigma\nu} + \partial_{\nu} g_{\sigma\mu} - \partial_{\sigma} g_{\mu\nu})$$

where $g^{\lambda\sigma}$ is the matrix inverse of $g_{\lambda\sigma}$ called the inverse metric. This is the solution of the relation (8.19) and the notation for the inverse metric is standard [cf (20.17)]. The components of the geodesic equation are

$$du^{\alpha}/d\tau = -\Gamma^{\alpha}_{\beta\gamma} u^{\beta} u^{\gamma}.$$

You must input the covariant components of the metric tensor $g_{\mu\nu}$ by editing the relevant input line in this *Mathematica* notebook. You may also wish to change the names of the coordinates. The nonzero components of the above quantities are displayed as the output.

■ Clearing the values of symbols:

First clear any values that may already have been assigned to the names of the various objects to be calculated. The names of the coordinates that you will use are also cleared.

```
Clear[coord, metric, inversemetric, affine, r,  $\theta$ ,  $\phi$ , t]
```

■ Setting The Dimension

The dimension n of the spacetime (or space) must be set:

```
n = 4
```

```
4
```

■ Defining a list of coordinates:

The example given here is the wormhole metric (7.40). Note that for convenience t is denoted by x^4 rather than x^0 and summations run from 1 to 4 rather than 0 to 3.

```
coord = {r,  $\theta$ ,  $\phi$ , t}
```

```
{r,  $\theta$ ,  $\phi$ , t}
```

You can change the names of the coordinates by simply editing the definition of **coord**, for example, to **coord = {x, y, z, t}**, when another set of coordinate names is more appropriate.

■ Defining the metric:

Input the metric as a list of lists, i.e., as a matrix. You can input the components of any metric here, but you must specify them as explicit functions of the coordinates.

```
metric = {{1, 0, 0, 0}, {0, r^2 + b^2, 0, 0}, {0, 0, (r^2 + b^2) Sin[θ]^2, 0}, {0, 0, 0, -1}}
{{1, 0, 0, 0}, {0, b^2 + r^2, 0, 0}, {0, 0, (b^2 + r^2) Sin[θ]^2, 0}, {0, 0, 0, -1}}
```

You can also display this in matrix form:

```
metric // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & b^2 + r^2 & 0 & 0 \\ 0 & 0 & (b^2 + r^2) \sin^2[\theta] & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

■ Note:

It is important not to use the symbols, **i**, **j**, **k**, **l**, **n**, or **s** as constants or coordinates in the metric that you specify above. The reason is that the first four of those symbols are used as summation or table indices in the calculations done below. The last is the dimension of the space.

■ Calculating the inverse metric:

The inverse metric is obtained through matrix inversion.

```
inversemetric = Simplify[Inverse[metric]]
```

$$\left\{ \{1, 0, 0, 0\}, \left\{0, \frac{1}{b^2 + r^2}, 0, 0\right\}, \left\{0, 0, \frac{\text{Csc}[\theta]^2}{b^2 + r^2}, 0\right\}, \{0, 0, 0, -1\} \right\}$$

The inverse metric can also be displayed in matrix form:

```
inversemetric // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \frac{1}{b^2 + r^2} & 0 & 0 \\ 0 & 0 & \frac{\text{Csc}[\theta]^2}{b^2 + r^2} & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

■ Calculating the affine connection:

The calculation of the components of the affine connection is done by transcribing the definition given earlier into the notation of *Mathematica* and using the *Mathematica* functions **D** for taking partial derivatives, **Sum** for summing over repeated indices, **Table** for forming a list of components, and **Simplify** for simplifying the result.

```
affine := affine = Simplify[Table[(1/2) * Sum[(inversemetric[[i, s]] *
  (D[metric[[s, j]], coord[[k]]] +
  D[metric[[s, k]], coord[[j]]] - D[metric[[j, k]], coord[[s]]]), {s, 1, n}],
  {i, 1, n}, {j, 1, n}, {k, 1, n}]]
```

■ Displaying the affine connection:

The nonzero components of the affine connections are displayed below. You need not follow the details of constructing the functions that we use for that purpose. Because the affine connection is symmetric under interchange of the last two indices, only the independent components are displayed.

```
listaffine := Table[If[UnsameQ[affine[[i, j, k]], 0],
  {ToString[Γ[i, j, k]], affine[[i, j, k]]}], {i, 1, n}, {j, 1, n}, {k, 1, j}]

TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 2], TableSpacing → {2, 2}]

Γ[1, 2, 2]    -r
Γ[1, 3, 3]    -r Sin[θ]2
Γ[2, 2, 1]     $\frac{r}{b^2+r^2}$ 
Γ[2, 3, 3]    -Cos[θ] Sin[θ]
Γ[3, 3, 1]     $\frac{r}{b^2+r^2}$ 
Γ[3, 3, 2]    Cot[θ]
```

■ Calculating the geodesic equations:

The geodesic equations are calculated by asking *Mathematica* to carry out the sum $-\Gamma^\alpha_{\beta\gamma} u^\beta u^\gamma$, where u^α are the components of the four-velocity. (This gives the derivative of u^α with respect to proper time τ . (This is replaced by s if the geodesics are spacelike.)

```
geodesic := geodesic = Simplify[Table[-Sum[affine[[i, j, k]] u[j] u[k], {j, 1, n},
  {k, 1, n}], {i, 1, n}]]
```

■ Displaying the geodesic equations:

```
listgeodesic := Table[{"d/dτ" ToString[u[i]], "=", geodesic[[i]]}, {i, 1, n}]

TableForm[listgeodesic, TableSpacing → {2}]

d/dτ u[1] = r (u[2]2 + Sin[θ]2 u[3]2)
d/dτ u[2] = - $\frac{2r u[1] u[2]}{b^2+r^2}$  + Cos[θ] Sin[θ] u[3]2
d/dτ u[3] = - $\frac{2r u[1] u[3]}{b^2+r^2}$  - 2 Cot[θ] u[2] u[3]
d/dτ u[4] = 0
```

■ Acknowledgment

This program was adapted from the notebook *Curvature and the Einstein equation* kindly written by *Leonard Parker* especially for this text.